Algebrinės operacijos su MAPLE

Sumos, atimties, sandaugos, dalybos operacijoms žymėti naudojami simboliai $+$, **-**, *****, $/$ o kėlimą laipsniu žymime simboliu `^`.

```
> (z+a*y-x)^3;
```

$$(z + a\,y - x)^3$$

Simboliui **r** suteikiame reiškinio reikšmę naudodami priskyrimo ženklą **:=**. Jei komanda užbaigsime simboliu **:** (o ne **;**) , ji bus įvykdyta, o rezultatas nebus spausdinamas ekrane:

```
> r:=(z+a*y-x)^3:
```

Priskirtąjį simboliui reiškinį galime atspausdinti ekrane naudodami komandą **print**:

```
> print(r);
```

$$(z + a\,y - x)^3$$

Algebrinį reiškinį reiškinį $r := (z + a\,y - x)^3$ išskleisime naudodami komandą **expand:**

```
> r1:=expand(r);
```

$$r1 := z^3 + 3\,z^2\,a\,y - 3\,x\,z^2 + 3\,z\,a^2\,y^2 - 6\,z\,a\,y\,x + 3\,z\,x^2 + a^3\,y^3 - 3\,a^2\,y^2\,x + 3\,a\,y\,x^2 - x^3$$

o norimu būdu surūšiuosime komandos **sort** pagalba:

```
> sort(r1);
```

$$a^3\,y^3 - 3\,a^2\,x\,y^2 + 3\,a^2\,y^2\,z + 3\,a\,x^2\,y - 6\,a\,x\,y\,z + 3\,a\,y\,z^2 - x^3 + 3\,x^2\,z - 3\,x\,z^2 + z^3$$

Simboliu **%** žymimas paskutinis skaičiavimų rezultatas. Jei norime surūšiuoti reiškinį pagal kintamojo **x** laipsnius, ra šome tokią komandą:

```
> sort(%,[x]);
```

$$-x^3 + 3\,a\,y\,x^2 + 3\,z\,x^2 - 6\,a\,y\,z\,x - 3\,a^2\,y^2\,x - 3\,z^2\,x + 3\,a\,y\,z^2 + 3\,a^2\,y^2\,z + a^3\,y^3 + z^3$$

Nurodę papildomą rūšiavimo tvarką, gausime kitokį rezultatą:

```
> sort(%,[x,y]);
```

$$-x^3 + 3\,a\,x^2\,y - 3\,a^2\,x\,y^2 + a^3\,y^3 + 3\,z\,x^2 - 6\,a\,z\,x\,y + 3\,a^2\,z\,y^2 - 3\,z^2\,x + 3\,a\,z^2\,y + z^3$$

Kartais patogu skaičiavimo rezultatus norisi pakomentuoti arba tiesiog įvardinti. Tai galime padaryti tarp pasvirųjų kabučių **'... '**:

```
> 'skleidinys pagal x,y,z'=sort(%,[x,y,z]);
```

$$\textit{skleidinys pagal } x, y, z = -x^3 + 3\,a\,x^2\,y + 3\,x^2\,z - 3\,a^2\,x\,y^2 - 6\,a\,x\,y\,z - 3\,x\,z^2 + a^3\,y^3 + 3\,a^2\,y^2\,z + 3\,a\,y\,z^2 + z^3$$

Šiam reiškinį sugrupuoti kintamųjų laipsniais naudojame funkciją **collect**:

```
> collect(r,x);
```

$$-x^3 + (3\,z + 3\,a\,y)\,x^2 + ((z + a\,y)\,(-2\,z - 2\,a\,y) - (z + a\,y)^2)\,x + (z + a\,y)^3$$

Komandą **factor** naudojame reiškiniui išskaidyti dauginamaisiais:

```
> factor(%);
```

$$(z + a\,y - x)^3$$

Jei norime paskaičiuoti reiškinį, kai reiškinio kintamieji dydžiai arba parametrai įgyja tam tikras reikšmes, naudojame operatorių **subs:**

```
> subs(x=2,a=1,r);
```

$$(z + y - 2)^3$$

Kai norime reiškinį pakomentuoti, galime įrašyti komentarą komandos eilutėje atskirdami jį ženklu #:

```
> a:=2:x:=1:y:=3:z:=5:r;#r reikšmė taške (1,3,5),a=2
```

$$1000$$

Pastebėkime, kad naudojant komandą **subs**, kintamieji **a, x, y, z** liko laisvi, o dabar kintamiesiems buvo negrįžtamai priskirtos skaitinės reikšmės:

```
> print(r1);
```

$$1000$$

Norėdami atlaisvinti kintamųjų reikšmes, naudojame komandą **restart**. Turėkime omenyje, kad ji panaikina visus ankstesniuosius skaičiavimų rezultatus.

```
> restart:
```

Reiškinio algebrinę išraišką galime pakeisti komandos convert **pagalba.** Keičiame dešimtainį skaičių trupmena:

```
> convert( 3.14,fraction );
```

$$\frac{157}{50}$$

Racionalųjį reiškinį

```
> f:=(x)-> (x^5+x^2+2)/(x^2-1);
```

$$f := x \to \frac{x^5 + x^2 + 2}{x^2 - 1}$$

galime išreikšti sveikosios dalies (daugianario) ir taisyklingųjų racionaliųjų reiškinių suma:

```
> convert(f(x), parfrac, x);
```

$$x^3 + x + 1 - \frac{1}{x+1} + \frac{2}{x-1}$$

Bet kurios komandos aprašymą gausite darbiniame lange surinkę po klaustuko ženklo **?** surinkę komandos vardą ir paspaudę klavišą enter. Pavyzdžiui, Atkreipkime dėmesį, kad reiškinį užrašėme kaip funkciją **f(x)**, naudodami atimties ir nelygybės simbolius **-** ir >.

```
> ?convert;
```

Atkreipkime dėmesį, kad reiškinį užrašėme kaip funkciją **f(x)**, naudodami atimties ir nelygybės simbolius **-** ir >. Tokiu atveju jos reikšmę, pavyzdžiui, taške $x := \pi$ gausime taip:

```
> f(Pi);
```

$$\frac{\pi^5 + \pi^2 + 2}{\pi^2 - 1}$$

Apytikslę dešimtainę reikšmę gausima komandos **evalf** pagalba:

```
> evalf(%);
```

$$35.84030075$$

Skaičiavimų tikslumą padidinsime nurodę norimą ženklų skaičių po kablelio:

```
> evalf(f(Pi),25);
```

$$35.84030074073290927925016$$

Apibrėžkime reiškinį:

```
> p:=(y^7-1)/(y-1);
```

$$p := \frac{y^7 - 1}{y - 1}$$

Komanda **simplify** suprastins reiškinį:

```
> s:=simplify(p);
```

$$s := y^6 + y^5 + y^4 + y^3 + y^2 + y + 1$$

Jį galime užrašyti kaip kintamojo **y** funkciją naudojant komand ą **unapply:**

```
> g(y):=unapply(s,y);
```

$$\mathrm{g}(y) := y \to y^6 + y^5 + y^4 + y^3 + y^2 + y + 1$$

Tą patį reiškinį galime užrašyti naudodami komandą **sum**:

```
> sum(y^i, i=0..6);
```

$$y^6 + y^5 + y^4 + y^3 + y^2 + y + 1$$

Funkcija **product** analogiškai veikia sandaugos atžvilgiu:

```
> product(k^k/(k+1), k=0..6 );
```

$$\frac{5598720000}{7}$$

Analogiškai veikia komandos **mul** ir **add**:

```
> mul( i^i/(i+1), i=0..6 );
> add( y^i, i=0..6 );
```

$$\frac{5598720000}{7}$$

$$y^6 + y^5 + y^4 + y^3 + y^2 + y + 1$$

Formulę

```
> (a-b)^2=a^2-2*a*b+b^2;
```

$$(a - b)^2 = a^2 - 2\,a\,b + b^2$$

galime patikrinti apskaičiavę jos kairiosios ir dešiniosios pusių skirtumą **lhs** ir **rhs** funkcijų pagalba:

```
> simplify(lhs(%)-rhs(%));
```

$$0$$

Užduotys.

1. Išskaidykite dauginamaisiais $x^6 - x^5 - 9\,x^4 + x^3 + 20\,x^2 + 12\,x$ reiškinį.

2. Patikrinkite Jums žinomų algebros formulių teisingumą.

3. Užrašykite reiškinį **n(n-1)..(n-k+1)** = **n!/(n-r)!** kaip kintamųjų k ir n funkciją ir apskaičiuokite jo reikšmes įvairioms argumentų reikšmėms.

2. Kompleksiniai skaičiai. Algebrinis, trigonometrinis ir rodiklinis kompleksinių skaičių pavidalas. Daugianarių šaknys.

Kompleksiniu skaičiumi vadinamas reiškinys $\boldsymbol{z = x + yi}$; čia x,y - realieji skaičiai, i - menamasis vienetas, turintis savybę $i^2 = -1$. Tokia kompleksinio skaičiaus forma vadinama jo algebrine forma. Skaičius x vadinamas kompleksinio skaičiaus z *realiaja dalimi* ir zymimas Re z , o skaicius y - menamaja dalimi ir zymimas Imz. Kompleksinio skaičių MAPLE užrašome su komanda **Complex:**

```
> z[1]:=Complex(2,5);
```

$$z_1 := 2 + 5\,I$$

Arba užrašome kaip algebrinį reiškinį naudodami **I** vietoje menamojo vieneto **i:**

```
> z[1] := 2 + 5*I;
```

$$z_1 := 2 + 5\,I$$

Realiąją kompleksinio skaičiaus dalį gausime su komanda **Re(z),** o menamąją - su **Im(z):**

```
>  z[1] := -3 + 7*I;
>  ‘Re(z[1])‘=Re(z[1]);
>  z[2]:=  9 + 4*I;
>  ‘Re(z[2])‘=Re(z[2]);;
```

$$z_1 := -3 + 7\,I$$

$$Re(z[1]) = -3$$

$$z_2 := 9 + 4\,I$$

$$Re(z[2]) = 9$$

```
>  ‘Im(z[1])‘=Im(z[1]);
>  ‘Im(z[2])‘=Im(z[2]);
```

$$Im(z[1]) = 7$$

$$Im(z[2]) = 4$$

```
>  x-I*y;
```

$$x - y\,I$$

Kompleksinio skaičiaus $x + y\,I$ jungtinis skaičius $x - I\,y$ randamas su komanda **conjugate**:

```
>  ‘jungtinis(z[1])‘ = conjugate(Z1);
>  ‘jungtinis(z2)‘ = conjugate(Z2);
```

$$jungtinis(z[1]) = \overline{Z1}$$

$$jungtinis(z2) = \overline{Z2}$$

Veiksmai su kompleksiniais skaičiais.

Kompleksiniu skaiciu z1=x1+y1*I ir z2=x2+y2*I **suma** vadinamas kompleksinis skaicius

z=x+y*I=z1+z2=(x1+x2)+(y1+y2)*I

Atitinkamai kompleksiniu skaiciu z1=x1+y1*I ir z2=x2+y2*I **skirtumu** vadinamas kompleksinis skaicius

z=x+y*I=z1-z2=(x1-x2)+(y1-y2)*I

Kompleksinių skaičių sumos, atimties, daugybos, dalybos veiksmams MAPLE naudojami įprasti simboliai:

```
>  z[1]:= 3 + 7*I;
>  z[2]:= 5 - 6*I;
>  ‘z[1] + z[2]‘ = z[1] + z[2];
>  ‘z[1] - z[2]‘ = z[1] - z[2];
```

$$z_1 := 3 + 7\,I$$
$$z_2 := 5 - 6\,I$$
$$z[1]\ +\ z[2] = 8 + I$$
$$z[1]\ -\ z[2] = -2 + 13\,I$$

```
> `z[1]*z[2]` = Z[1]*Z[2];
```

$$z[1] * z[2] = Z_1\,Z_2$$

```
> `z[1]/z[2]`=Z[1]/Z[2];
```

$$z[1]/z[2] = \frac{Z_1}{Z_2}$$

Kompleksinių reiškinių reikšmės apskaičiuojamos su komanda **evalc**:

```
> z:=(sqrt(5)+3*I)^2;
```

$$z := (\sqrt{5} + 3\,I)^2$$

```
> evalc(z);
```

$$-4 + 6\,I\,\sqrt{5}$$

Kompleksinio skaičiaus modulis apskaičiuojamas komandos **abs** pagalba:

```
> `z`=sqrt(2+5*I);
```

$$(\sqrt{5} + 3\,I)^2 = \sqrt{2 + 5\,I}$$

```
> evalc(z);
```

$$-4 + 6\,I\,\sqrt{5}$$

```
> `modz`=abs(z);
```

$$modz = 14$$

Bet kokiems kompleksiniams skaičiams teisinga **trikampio nelygybė** $|z_1 + z_2| \leq |z_1| + |z_2|$:

```
> z1:= 7 +I;
> z2:= 3 +5*I;
> z3:=-2 -7*I;
> `z1 + z2 +z3 `= z1 + z2 + z3;
> `|z1| ` = abs(z1);
> `|z2| ` = abs(z2);
> `|z3| ` = abs(z3);
> `|z1 + z2 + z3| ` = abs(z1 + z2 +z3); ` `;
> evalf(abs(z1 + z2 +z3) <= abs(z1) + abs(z2) + abs(z3));
```

$$z1 := 7 + I$$
$$z2 := 3 + 5\,I$$
$$z3 := -2 - 7\,I$$
$$z1\ +\ z2\ + z3\ = 8 - I$$
$$|z1|\ = 5\,\sqrt{2}$$
$$|z2|\ = \sqrt{34}$$
$$|z3|\ = \sqrt{53}$$

$$|z1 \ + \ z2 \ + \ z3| \ = \sqrt{65}$$

$$8.062257748 \leq 20.18212959$$

```
{\large Trigonometrinė (polinė) kompleksinių skaičių forma}
```

Tai išraiška, $z = \mathrm{r}(\cos(\theta) + i\sin(\theta))$, kur r - kompleksinio skaičiaus modulis, o θ - argumentas. Skaičiaus $z := -1 - I$ polinę formą galime užrašyti taip:

```
> z := -1 - I;
```

$$z := -1 - I$$

```
> `polz`:=abs(z)*(cos(theta)+I*sin(theta));
```

$$polz := \sqrt{2}\,(\cos(\theta) + \sin(\theta)\,I)$$

Argumentą galime gauti su komanda **argument:**

```
> `theta`:=argument(z);
```

$$\theta := -\frac{3\,\pi}{4}$$

Funkcija **polar** randa kompleksinio skaičiaus polinę formą tokiu pavidalu:

```
> `polarz`:=polar(z);
```

$$polarz := \mathrm{polar}(\sqrt{2},\, -\frac{3\,\pi}{4})$$

Sudauginkime ir padalinkime skaičius poline forma:

```
> z1:=1+2*I;
```

$$z1 := 1 + 2\,I$$

```
> polar(z1);
```

$$\mathrm{polar}(\sqrt{5},\, \arctan(2))$$

```
> `polar(z*z1)`:=polar(sqrt(2),-3/4*Pi)*polar(sqrt(5),arctan(2));
```

$$polar(z*z1) := \mathrm{polar}(\sqrt{2},\, -\frac{3\,\pi}{4})\,\mathrm{polar}(\sqrt{5},\, \arctan(2))$$

```
> simplify(%);
```

$$\mathrm{polar}(\sqrt{2}\,\sqrt{5},\, -\frac{3\,\pi}{4} + \arctan(2))$$

```
> `polar(z/z1)`:=polar(sqrt(2),-3/4*Pi)/polar(sqrt(5),arctan(2));
```

$$polar(z/z1) := \frac{\mathrm{polar}(\sqrt{2},\, -\frac{3\,\pi}{4})}{\mathrm{polar}(\sqrt{5},\, \arctan(2))}$$

```
> simplify(%);
```

$$\mathrm{polar}(\frac{\sqrt{2}\,\sqrt{5}}{5},\, -\frac{3\,\pi}{4} - \arctan(2))$$

Rodiklinė (eksponentinė) kompleksinio skaičiaus forma.

Užrašykime skaičių $z = 1{+}4$i rodikline forma $z = r\,e^{(i\,\theta)}$:

```
> z:=1+4*I;
```

$$z := 1 + 4\,I$$

```
> theta:=argument(z);
```

$$\theta := \arctan(4)$$

```
> r:=abs(z);
```

$$r := \sqrt{17}$$

```
> rodz:=r*exp(I*theta);
```

$$rodz := 1 + 4\,I$$

Su evalb patikriname išraiškos teisingumą:

```
> evalb(rodz=z);
```

$$true$$

Daugianarių šaknų radimas

Suraskime visas lygties $z^3 = 1$ šaknis ir pavaizduokime jas grafiskai.S

```
> restart:‘z^3 = 1 sprendiniai.‘; ‘ ‘;
> solset := {solve(z^8 = 1, z)}:
> ‘Sprendiniai ‘ = solset;
> pts := map(w->[Re(w),Im(w)], solset):
> plot(pts,
> style=point, symbol=circle,
> scaling=constrained, color=red,
> labels=[‘x‘,‘y ‘],
> view=[-1.1..1.1,-1.1..1.1]);
```

$$z\hat{}3\ =\ 1\ sprendiniai.$$

$$Sprendiniai\ = \{-1, 1, -I, I, -\frac{\sqrt{2}}{2} - \frac{1}{2}\,I\,\sqrt{2}, -\frac{\sqrt{2}}{2} + \frac{1}{2}\,I\,\sqrt{2}, \frac{\sqrt{2}}{2} - \frac{1}{2}\,I\,\sqrt{2}, \frac{\sqrt{2}}{2} + \frac{1}{2}\,I\,\sqrt{2}\}$$



Išspręskime lygtį $z^2 + (1+i)\,z + 5\,i = 0$:

```
> solset := {solve(z^2 +(1+I)*z +5*I, z)}:
> `Sprendiniai ` = solset;
> pts := map(w->[Re(w),Im(w)], solset):
> plot(pts,
> style=point, symbol=circle,
> scaling=constrained, color=red,
> labels=[`    x`,`y  `],
> view=[-2.1..2.1,-2.1..2.1]);
```

$$\mathit{Sprendiniai}\ = \{-2+I,\,1-2\,I\}$$



```
>  Kartais sprendinius gausime RootOff pavidalu:  tada
sprendinius apskaičiuosime apytiksliai naudodami komandą "evalf":
```

Jei sprendinius gauname RootOff pavidalu, juosn apytiksliai apskai čiuosime naudodami komandą **evalf**:

```
> solset := {solve(z^5 +(1+I)*z +5*I, z)}:
> ‘Sprendiniai‘ = solset;
> pts := map(w->[Re(w),Im(w)], solset):
> plot(pts,
> style=point, symbol=circle,
> scaling=constrained, color=red,
> labels=[‘    x‘,‘y  ‘],
> view=[-2.1..2.1,-2.1..2.1]);
```

$$\mathit{Sprendiniai} = \{\mathrm{RootOf}(\_Z^5 + (1+I)\,\_Z + 5\,I,\ \mathit{index} = 1),$$
$$\mathrm{RootOf}(\_Z^5 + (1+I)\,\_Z + 5\,I,\ \mathit{index} = 2),$$
$$\mathrm{RootOf}(\_Z^5 + (1+I)\,\_Z + 5\,I,\ \mathit{index} = 3),$$
$$\mathrm{RootOf}(\_Z^5 + (1+I)\,\_Z + 5\,I,\ \mathit{index} = 4),$$
$$\mathrm{RootOf}(\_Z^5 + (1+I)\,\_Z + 5\,I,\ \mathit{index} = 5)\}$$




```
> evalf(solset);
```

$\{-1.19872968985837 - 0.448712697377034\, I,$
$-0.908869507386465 + 1.07542529543075\, I,$
$-0.0850322880474402 - 1.30403624881022\, I,$
$0.857380430484718 + 1.20481621974755\, I,$
$1.33525105480755 - 0.527492568991053\, I\}$

Kompleksiniu skaiciu matricine interpretacija

```
> with(linalg) :
```

Vienetą algebrinėje kompleksinio skaičiaus išraiškoje keičiame vienetine matrica, o menamąjį vienetą - matrica M:

```
> M:=matrix(2,2,[ 0,-1,1,0]);
```

$$M := \begin{bmatrix} 0 & -1 \\ 1 & 0 \end{bmatrix}$$

```
> E:=Matrix(2,2,shape=identity);
```

$$E := \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

```
> multiply(M,M);
```

$$\begin{bmatrix} -1 & 0 \\ 0 & -1 \end{bmatrix}$$

Taigi, **M** elgiasi kaip menamasis vienetas. Pastoviosios matricos tapatinamos su vektorių erdvės tiesiniu atvaizdavimu:

```
> V:=vector([u,v]);
```

$$V := [u,\, v]$$

```
> multiply(E,V);
```

$$[u,\, v]$$

```
> multiply(M,V);
```

$$[-v,\, u]$$

Matome, kad **E** nekeičia vektoriaus, o **M** pasuka jį 90 laipsnių kampu prieš laikrodžio rodyklę.

Uždaviniai.

1. Suraskite kompleksinio skaičiaus (1-2*I)^3/(5+I) algebrinę formą, polinę ir rodiklinę formas, realiąją ir menamąją dalis, jungtinį kompleksinį skaičių: .

2. Suraskite lygties $z^3 - 3\,z + 3 = 0$ sprendinius ir atvaizduokite grafiškai.

3. Matricos, determinantai. Atvirkštinė matrica. Tiesinių lygčių sistemų sprendimas.

Matricos užrašymas MAPLE

MAPLE matricas galima užrašyti (užduoti, įvesti) keletu būdų. Apibūdinsime pagrindinius iš jų. Pirmiausiai aktyvuojame **linalg** paketą:

```
> with(linalg):
```

Galime užrašyti matricą išvardindami visų jos eilučių elementus:

```
> A:=matrix( [ [1,2,3],[2,8,5],[3,0,10] ] );
```

$$A := \begin{bmatrix} 1 & 2 & 3 \\ 2 & 8 & 5 \\ 3 & 0 & 10 \end{bmatrix}$$

Arba išvardindami visus matricos elementus paeiliui, prieš tai nurodę matricos formatą:

```
> A:=matrix(3,3,[1,2,3,2,8,5,3,0,10]);
```

$$A := \begin{bmatrix} 1 & 2 & 3 \\ 2 & 8 & 5 \\ 3 & 0 & 10 \end{bmatrix}$$

Galime apibrėžti matricos elementus kaip tų elementų indeksų funkcijas, prieš tai nurodę matricos formatą:

```
> B:=matrix(3,3,(i,j)->(i+j)^j);
```

$$B := \begin{bmatrix} 2 & 9 & 64 \\ 3 & 16 & 125 \\ 4 & 25 & 216 \end{bmatrix}$$

Jei norime esamoje matricoje pakeisti vieną jos elementą, tai galime padaryti šitaip:

```
> A[2,3]:=-8:  A=evalm(A);
```

$$A = \begin{bmatrix} 1 & 2 & 3 \\ 2 & 8 & -8 \\ 3 & 0 & 10 \end{bmatrix}$$

Vienetinę matricą patogu apibrėžti identity komandos pagalba:

```
> E:=Matrix(4,4,shape=identity);
```

$$E := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

PASTABA: Žodis **Matrix** komandoje būtinai rašomas didžiąją raide.

Matricas, kurių visi elementai vienodi, patogu užrašyti šitaip:

```
> A:=matrix(2,5,1);
```

$$A := \begin{bmatrix} 1 & 1 & 1 & 1 & 1 \\ 1 & 1 & 1 & 1 & 1 \end{bmatrix}$$

Matricą, kurios elementai atsitiktiniai skaičiai, galime generuoti komandos **randmatrix** pagalba:

```
> B:=randmatrix(2,5);
```

$$B := \begin{bmatrix} -94 & 87 & -56 & 0 & -62 \\ 97 & -73 & -4 & -83 & -10 \end{bmatrix}$$

Sklaisteliuose nurodėme matricos formatą.

Matricų veiksmai

Matricų suma užrašoma taip pat, kaip skaičių suma:

```
> C:=A+B;
```

$$C := A + B$$

Komanda evalm suskaičiuoja matricos elementų reikšmes (analogiškai, kaip komanda evalf - skaliarinių dydžių reikšmes):

```
> F:=evalm(%);
```

$$F := \begin{bmatrix} -93 & 88 & -55 & 1 & -61 \\ 98 & -72 & -3 & -82 & -9 \end{bmatrix}$$

Transponuota matrica randama komandos transpose pagalba:

```
> Btr:=transpose(B);
```

$$Btr := \begin{bmatrix} -94 & 97 \\ 87 & -73 \\ -56 & -4 \\ 0 & -83 \\ -62 & -10 \end{bmatrix}$$

Matricos rangą apskaičiuojame naudodami rank komandą:

```
> rank(B);
```

$$2$$

```
> rank(A);
```

$$1$$

Raskime atsitiktinės matricos trečiąjį laipsnį:

```
> A:=randmatrix(5,5);
```

$$A := \begin{bmatrix} 62 & -82 & 80 & -44 & 71 \\ -17 & -75 & -10 & -7 & -40 \\ 42 & -50 & 23 & 75 & -92 \\ 6 & 74 & 72 & 37 & -23 \\ 87 & 44 & 29 & 98 & -23 \end{bmatrix}$$

```
> `A^3`=evalm(A^3);
```

$$A\hat{}3 = \begin{bmatrix} 1254309 & -619634 & 1993294 & 181513 & 349738 \\ -116825 & -66361 & -560888 & 175777 & -246637 \\ 340863 & -450140 & -432072 & 764033 & -840555 \\ -503026 & 253734 & 179775 & -496828 & 162154 \\ 940656 & -663012 & 683056 & 842423 & -694216 \end{bmatrix}$$

Naudojome tą patį kėlimo laipsniu simbolį `^`, kurį naudojame skaliarinių dydžių atveju. Bet kurias suderinamas matricas dauginsime naudodami simbolį &* :

```
> G:=evalm(F&*A);
```

$$G := \begin{bmatrix} -14873 & 1166 & -11282 & -6590 & -3683 \\ 5899 & -8950 & 2326 & -7949 & 12207 \end{bmatrix}$$

arba komandą multiply:

```
> multiply(F, A);
```

$$\begin{bmatrix} -14873 & 1166 & -11282 & -6590 & -3683 \\ 5899 & -8950 & 2326 & -7949 & 12207 \end{bmatrix}$$

multiply komanda leidžia iškart sudauginti ir didesnį negu 2 dauginamųjų skai čių:

```
> multiply(F, A, transpose(G));
```

$$\begin{bmatrix} 406841798 & -116987930 \\ -116987930 & 332508427 \end{bmatrix}$$

Matricą dauginame iš skaičiaus naudodami įprastą daugybos simbolį *:

```
> ‘5*A‘:=evalm(5*A);
```

$$5*A := \begin{bmatrix} 310 & -410 & 400 & -220 & 355 \\ -85 & -375 & -50 & -35 & -200 \\ 210 & -250 & 115 & 375 & -460 \\ 30 & 370 & 360 & 185 & -115 \\ 435 & 220 & 145 & 490 & -115 \end{bmatrix}$$

Kvadratinės matricos determinantą apskaičiuojame komandos det pagalba:

```
> ‘Matricos A determinantas‘:=det(A);
```

$$\textit{Matricos A determinantas} := 769100325$$

Atvirkštinę matricą randame suinverse komanda:

```
> ‘Aatv‘:=inverse(A);
```

$$Aatv := \begin{bmatrix} \frac{6619832}{769100325} & \frac{-21625328}{153820065} & \frac{3368727}{36623825} & \frac{-5012519}{153820065} & \frac{-7061287}{109871475} \\ \frac{-31103}{769100325} & \frac{-10215103}{153820065} & \frac{1424117}{36623825} & \frac{-1396114}{153820065} & \frac{-3416327}{109871475} \\ \frac{488357}{153820065} & \frac{894694}{30764013} & \frac{-122173}{7324765} & \frac{478057}{30764013} & \frac{228548}{21974295} \\ \frac{-7084307}{769100325} & \frac{26845208}{153820065} & \frac{-4203452}{36623825} & \frac{5132669}{153820065} & \frac{10303012}{109871475} \\ \frac{-708601}{256366775} & \frac{6227434}{51273355} & \frac{-3213608}{36623825} & \frac{1084082}{51273355} & \frac{2439316}{36623825} \end{bmatrix}$$

Nepamirškime, kad tik reguliari matrica (tokia, kurios determinantas nelygus nuliui) turi atvirkštinę. Patikrinsime, ar tikrai gautoji matrica Aatv yra matricos A atvirkštinė matrica:

```
> evalm(Aatv&*A);
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

Vektoriai

Vektoriams užrašyti galime naudoti komandą vector:

```
> vector( [5,4,3,2] );
```

$$[5, 4, 3, 2]$$

Jei visos vektoriaus komponentės vienodos, galime nurodyti tik vektoriaus matavimą ir komponenčių vertę:

```
> vector(5, 7);
```

$$[7, 7, 7, 7, 7]$$

Galime vektoriaus komponentes užrašyti kaip komponenčių indeksų funkcijas:

```
> f := x -> 6-x:
> v := vector(4, f);
```

$$v := [5, 4, 3, 2]$$

Matome, kad šitokiu būdu ušrašėme pirmąjį nagrinėtąjį vektorių. Norimas vektoriaus komponentes galime gauti taip:

```
> v[1],v[4];
```

$$5, 2$$

Tiesinės lygčių sistemos

Panagrinėkime keletą tiesinių lygčių sistemų sprendimo būdų. Tegul pavyzdžiu būna sistema

5x + 8y - 3z = 21

3x + 11y +4z = 13

x + 5y - 5z = 12

Sistemos matrica A yra:

```
> A:=matrix([ [5,8,-3],[3,11,4],[1,5,-5]]);
```

$$A := \begin{bmatrix} 5 & 8 & -3 \\ 3 & 11 & 4 \\ 1 & 5 & -5 \end{bmatrix}$$

Dešiniųjų lygčių pusių reikšmių vektorius B toks:

```
> B:=[21,13,12];
```

$$B := [21, 13, 12]$$

Nežinomų kintamųjų vektorius X:

```
> X:=[x,y,z];
```

$$X := [x,\ y,\ z]$$

Sistemą galima užrašyti matriciniu pavidalu AX=B:

```
> evalm(A&*X)=B;
```

$$[5\,x+8\,y-3\,z,\ 3\,x+11\,y+4\,z,\ x+5\,y-5\,z]=[21,\ 13,\ 12]$$

Panagrinėkime tris tiesinių lygčių sistemų sprendimo būdus.

I būdas. Naudojame komandą linsolve, kuri sprendžia sistemą AX=B:

```
> X:=linsolve(A,B);
```

$$X := [2,\ 1,\ -1]$$

II būdas. Sistemą sprendžiame atvirkštinės matricos metodu, sprendinį gauname pavidalu X=A^(-1)&*B:

```
> Atv:=inverse(A);
```

$$Atv := \begin{bmatrix} \frac{15}{47} & \frac{-5}{47} & \frac{-13}{47} \\ \frac{-19}{235} & \frac{22}{235} & \frac{29}{235} \\ \frac{-4}{235} & \frac{17}{235} & \frac{-31}{235} \end{bmatrix}$$

```
> X:=evalm(Atv &* B);
```

$$X := [2,\ 1,\ -1]$$

III būdas. Sistemą sprendžiame Gauso ir Žordano metodu. Sudarome išplėstąją sistemos matricą, naudodami komandą augment:

```
> AB:=augment(A,B);
```

$$AB := \begin{bmatrix} 5 & 8 & -3 & 21 \\ 3 & 11 & 4 & 13 \\ 1 & 5 & -5 & 12 \end{bmatrix}$$

ir naudojame komandą gaussjord:

```
> G_J:=gaussjord(AB);
```

$$G\_J := \begin{bmatrix} 1 & 0 & 0 & 2 \\ 0 & 1 & 0 & 1 \\ 0 & 0 & 1 & -1 \end{bmatrix}$$

Sistemos sprendinio **X** komponentes matome paskutiniame šios matricos stulpelyje. Akivaizdumo dėlei stackmatrix komandos pagalba galime šią matricą papildyti nežinomųjų eilute:

```
> GJ:=stackmatrix([[x,y,z,b]],G_J);
```

$$GJ := \begin{bmatrix} x & y & z & b \\ 1 & 0 & 0 & 2 \\ 0 & 1 & 0 & 1 \\ 0 & 0 & 1 & -1 \end{bmatrix}$$

Tada turime raidėmis x,y,z pažymėtus atitinkamus nežinomųjų koeficientų stulpelius. Tada akivaizdžiai matome, kad sistema, ekvivalenti pradinei, kurios atrodo taip:

x = 2, y = 1,z = -1 (jos **Gauso-Žordano** matrica pavadinta GJ).

Kitas pavyzdys:

```
> with(linalg):
> A:=matrix( [ [1,-1,1,1],[2,-1,-1,1],[1,-2,4,2] ] );
```

$$A := \begin{bmatrix} 1 & -1 & 1 & 1 \\ 2 & -1 & -1 & 1 \\ 1 & -2 & 4 & 2 \end{bmatrix}$$

```
> B:=[2,3,1];
```

$$B := [2,\ 3,\ 1]$$

```
> AB:=augment(A,B);
```

$$AB := \begin{bmatrix} 1 & -1 & 1 & 1 & 2 \\ 2 & -1 & -1 & 1 & 3 \\ 1 & -2 & 4 & 2 & 1 \end{bmatrix}$$

```
> G_J:=gaussjord(AB);
```

$$G\_J := \begin{bmatrix} 1 & 0 & -2 & 0 & 0 \\ 0 & 1 & -3 & -1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

Paskutinė matricos eilutė atitinka lygtį 0*x1+0*x2+0*x3+0*x4=1. Ši lygybė negalima, vadinasi, sistema sprendinių neturi.

Dar vienas uždavinys: raskite visus lygčių sistemos

x1 + 2*x2 + 1*x3 + 2*x4 - x5 = 2,

3*x1 - 2*x2 - 5*x4 - x5 = -2,

-2*x1 + 3x2 + *x3 - 4*x4 = 3,

sprendinius.

```
> S:=matrix( [ [1,2,1,2,-1],[3,-2,0,-5,-1],[-2,3,1,-4,0] ] );
```

$$S := \begin{bmatrix} 1 & 2 & 1 & 2 & -1 \\ 3 & -2 & 0 & -5 & -1 \\ -2 & 3 & 1 & -4 & 0 \end{bmatrix}$$

```
> B:=[2,-2,3];
```

$$B := [2,\ -2,\ 3]$$

```
> X:= linsolve(S,B);
```

$$X := [\_t_1,\ 1 - 11\,\_t_2,\ 2\,\_t_1 + 37\,\_t_2,\ \_t_2,\ 3\,\_t_1 + 17\,\_t_2]$$

čia t1, t2 - bet kokie realūs skaičiai. Patikrinkime, ar gautasis vektorius X tikrai tenkina sistem:

```
> Patikrinimas:=evalm (S &* X)=B;
```

$$Patikrinimas := [2,\ -2,\ 3] = [2,\ -2,\ 3]$$

Kaip matome, X tikrai yra duotosios sistemos sprendinys. Kadangi sistema turi be galo daug sprendinių, **X** vadiname - **bendruoju sistemos sprendiniu**.

Uždaviniai

1. Apskaičiuokite matricos $A := \begin{bmatrix} -5 & -7 & 6 & 1 \\ 4 & -3 & 2 & 2 \\ 6 & 6 & -4 & -6 \\ 4 & 5 & 5 & 3 \end{bmatrix}$ determinantą.

Raskite atvirkštinę matricą. Patikrinkite rezultatą.

2. Duota matrica $A := \begin{bmatrix} -1 & -3 & -1 \\ -2 & 3 & -3 \\ -1 & 5 & -1 \end{bmatrix}$ ir vektorius $B := [-13,\ 8,\ 19]$.

Išspręskite lygčių sistemą AX = B įvairiais būdais. Jei sprendinys vienintelis, pakeiskite uždavinio sąlygą taip, kad sistema turėtų be galo daug sprendinių.

4. Kvadratinės formos

n kintamųjų x1,x2 .... xn **kvadratine forma** vadiname reiškinį:

Q(x1, x2, ... , xn) = sum(aij*xi*xj, i,j=1..n),

(aij = aji),i,j = 1,...,n

Akivaizdu, kad kvadratinė forma vienareikšmiškai apibrėžiama **simetriška** (aij = aji) matrica

A=(aij), i,j=1..n.

Matricos simetriškumo sąlygą galime užrašyti tokiu pavidalu:

aij = (aij + aji)/2,

Jei **x** pažymėsime vektorių x=(x1, x2, ... ,xn), o x' - jo transponuotąjį vektorių, tai kvadratinę formą galime užrašyti taip:

Q(x) = x'Ax,

čia simbolis &* žymi matricų daugybą. Arba

Q(x) = (x,Ax),

čia skliaustai reiškia juose esančių dviejų vektorių: x ir A&*x skaliarinę sandaugą.

Simetrišką matricą A diagonalizuoja **ortogonalioji** matrica P, t.y. tokia kad jos atvirkštinė matrica lygi jos pačios transponuotajai matricai:

P'P=E,

kurE - vienetinė matrica. Taigi, jei pažymėsime DA matricos **A** diagonalizuotąją matricą (t.y. jos Žordano formą), tai

DA=P'AP,

o matricos

D=(dij), i=1..n,

elementai, kurie nėra pagrindinėje įstrižainėje, lygūs nuliui:

dij=o, i=/j.

Tiesiškai transformuojant erdvę matricos P pagalba: x=Py, kur y=(y1,y2,...,yn), kvadratinė forma Q(x) įgyja diagonalinį pavidalą DQ(y):

DQ(y):=y'DAy=(y,DAy).

Tada, suprantama, ji lygi:

DQ(y)=sum(dii*yi^2); i=1..n,

ir dalis dėmenų šioje sumoje gali buti lygūs nuliui.

Nenulinių dėmenų diagonaliniame (kanoniniame) kvadratinės formos Q(x) pavidale DQ(y) skaičius vadinamas kvadratinės formos Q(x) rangu. Teigiamų dėmenų skaičius vadinamas formos **signatūra**. Forma Q(x) vadinama **teigiamai (neigiamai) apibrėžta** , jei visi matriccos **DA** elementai t**eigiami (neigiami).** Kitais atvejais sakome, kad Q(x) yra **neapibrėžto ženklo** kvadratinė forma.

Taigi, kadangi

$$(x,Ax)=Q(x)=DQ(y)=(y,DAy), \text{ jei} x=Py,$$

teigiamas formosQ(x) apibrėžtumas reiškia, kad bet kokiam nenuliniam vektoriui x skaliarinė sandauga

$$(x,Ax)$$

nelygi 0.

```
> restart;
> with(linalg): with(plots): with(plottools):
```

1. Tarkim, turime dviejų kintamųjų kvadratinę formą Q(x)= 3*x[1]^2 - 2*x[2]^2 -12*x[1]*x[2]:

```
> Q:=3*x[1]^2-2*x[2]^2-12*x[1]*x[2];X:=[x[1],x[2]]; N:=nops(X);
```

$$Q := 3\,x_1{}^2 - 2\,x_2{}^2 - 12\,x_1\,x_2$$
$$X := [x_1,\, x_2]$$
$$N := 2$$

Sudarome šitos kvadratinės formos matricą:

```
> A:= matrix(2,2,[[3,-6],[-6,-2]]);
```

$$A := \begin{bmatrix} 3 & -6 \\ -6 & -2 \end{bmatrix}$$

Rasime jos **tikrinius vektorius x** ir **tikrines reikšmes** λ (tenkinančius lygtį $\mathbf{Ax} = \lambda\mathbf{x}$)

```
> lambda;
```

$$\lambda$$

```
> Tikrinis_vektorius := [eigenvectors(A)];
```

$$\mathit{Tikrinis\_vektorius} := [[-6,\, 1,\, \{\left[1,\, \frac{3}{2}\right]\}],\, [7,\, 1,\, \{\left[\frac{-3}{2},\, 1\right]\}]]$$

Normuojame tikrinius vektorius (pakeičiame juos lygiagrečiais pradiniams vienetinio ilgio vektoriais) **normalize** komanda :

```
> v1 := normalize(Tikrinis_vektorius[1][3][1]);
> v2 := normalize(Tikrinis_vektorius[2][3][1]);
```

$$\mathit{v1} := \left[\frac{\sqrt{13}\,\sqrt{4}}{13},\, \frac{3\,\sqrt{13}\,\sqrt{4}}{26}\right]$$

$$v2 := \left[-\frac{3\sqrt{13}\sqrt{4}}{26}, \frac{\sqrt{13}\sqrt{4}}{13}\right]$$

augment komandos pagalba konstruojame matricą P, kuri diagonalizuoja A:

```
> PA := augment(v1,v2);
```

$$PA := \begin{bmatrix} \frac{\sqrt{13}\sqrt{4}}{13} & -\frac{3\sqrt{13}\sqrt{4}}{26} \\ \frac{3\sqrt{13}\sqrt{4}}{26} & \frac{\sqrt{13}\sqrt{4}}{13} \end{bmatrix}$$

Tada P'AP=DA:

```
> DA := simplify(evalm(transpose(PA) &* A &* PA));
```

$$DA := \begin{bmatrix} -6 & 0 \\ 0 & 7 \end{bmatrix}$$

Įsitikiname, kad PA - ortogonali matrica:

```
> evalm(transpose(PA)&*PA):simplify(%);
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

Randame diagonalizuotąją kvadratinę formą:

```
> Y:=[y[1],y[2]];
```

$$Y := [y_1, y_2]$$

```
> DQ:=evalm(transpose(Y)&*DA&*Y);
```

$$DQ := -6\,{y_1}^2 + 7\,{y_2}^2$$

Akivaizdu, kad kvadratinė forma yra neapibrėžto ženklo, jos rangas 2, o signatūra lygi 1. Patikrinkime, ar keitinys X=**PAY** kanonizuoja kvadratinę formą **Q:**

```
> evalm(X=PA&*Y);
```

$$[x_1, x_2] = \left[\frac{1}{13}\sqrt{13}\sqrt{4}\,y_1 - \frac{3}{26}\sqrt{13}\sqrt{4}\,y_2, \frac{3}{26}\sqrt{13}\sqrt{4}\,y_1 + \frac{1}{13}\sqrt{13}\sqrt{4}\,y_2\right]$$

```
> K:=[seq(x[k]=rhs(%)[k],k=1..nops(X))];
```

$$K := [x_1 = \frac{1}{13}\sqrt{13}\sqrt{4}\,y_1 - \frac{3}{26}\sqrt{13}\sqrt{4}\,y_2, x_2 = \frac{3}{26}\sqrt{13}\sqrt{4}\,y_1 + \frac{1}{13}\sqrt{13}\sqrt{4}\,y_2]$$

```
> subs(K,Q):expand(%);
```

$$-6\,{y_1}^2 + 7\,{y_2}^2$$

Naudodami **implicitplot** nubrėžkime ir palyginkime kreives **Q** = **1** ir **DQ** =**1:**

```
> implicitplot(Q=1,x[1]=-2..2,x[2]=-2..2,scaling=constrained);
```




```
> implicitplot(DQ=1,y[1]=-2..2,y[2]=-2..2,scaling=constrained);
```



Tokiu būdu galime įsivaizduoti kaip tiesinė transformacija $\mathbf{X} = \mathbf{P\&{*}Y}$ paveikė erdvę.

2. Tarkime, jog turime trijų kintamųjų kvadratinę formą:

```
> restart:with(linalg):
> Q:=x[1]^2+x[2]^2+5*x[3]^2-6*x[1]*x[2]-2*x[1]*x[3]+2*x[2]*x[3];
```

$$Q := {x_1}^2 + {x_2}^2 + 5\,{x_3}^2 - 6\,x_1\,x_2 - 2\,x_1\,x_3 + 2\,x_2\,x_3$$

Pažymime kintamuosius

```
> indets(Q):X:=sort(convert(%,list));N:=nops(%);
```

$$X := [x_1,\, x_2,\, x_3]$$
$$N := 3$$

ir randame kvadratinės formos matricą **A**:

```
> A := hessian(Q/2,X);
```

$$A := \begin{bmatrix} 1 & -3 & -1 \\ -3 & 1 & 1 \\ -1 & 1 & 5 \end{bmatrix}$$

Kadangi simetrinė matrica **A** diagonalizuojama, jos diagonalusis pavidalas sutampa su jos **Žordano** pavidalu. Todėl galime kvadratinės formos matricos **A** diagonaliojo pavidalo **DA** ir diagonalizuojančiosios matricos **P** ieškoti naudodami **jordan** komandą:

```
> DA := jordan(A,R);
```

$$DA := \begin{bmatrix} -2 & 0 & 0 \\ 0 & 3 & 0 \\ 0 & 0 & 6 \end{bmatrix}$$

Jei mums reikia tik kanoninio kvadratinės formos **Q** pavidalo, jį galime gauti taip:

```
> Y:=[y||(1..nops(X))];evalm(transpose(Y)&*DA&*Y);
```

$$Y := [y1,\, y2,\, y3]$$
$$-2\,y1^2 + 3\,y2^2 + 6\,y3^2$$

Jei mums reikia kvadratinės formos kintamųjų keitinio, kuris kanonizuoja formą, randame diagonalizuojančią matricą **R**:

```
> print(R);
```

$$\begin{bmatrix} \frac{1}{2} & \frac{1}{3} & \frac{1}{6} \\ \frac{1}{2} & \frac{-1}{3} & \frac{-1}{6} \\ 0 & \frac{1}{3} & \frac{-1}{3} \end{bmatrix}$$

Naudodami **col** komandą išskiriame matricos **R** vektorius-stulpelius:

```
> v:=[col(R,1..N)];
```

$$v := \left[\left[\frac{1}{2},\, \frac{1}{2},\, 0\right],\, \left[\frac{1}{3},\, \frac{-1}{3},\, \frac{1}{3}\right],\, \left[\frac{1}{6},\, \frac{-1}{6},\, \frac{-1}{3}\right]\right]$$

GramSchmidt ir **normalize** komandomis randame ortonormuotąją vektorių erdvės bazę:

```
> GramSchmidt(v,normalized);
```

$$[\left[\frac{\sqrt{2}}{2},\, \frac{\sqrt{2}}{2},\, 0\right],\, \left[\frac{\sqrt{3}}{3},\, -\frac{\sqrt{3}}{3},\, \frac{\sqrt{3}}{3}\right],\, \left[\frac{\sqrt{6}}{6},\, -\frac{\sqrt{6}}{6},\, -\frac{\sqrt{6}}{3}\right]]$$

ir iš bazės vektorių sudarome ortogonaliąją matricą **P**:

```
> P:=transpose(matrix(N,N,%));orthog(P);
```

$$P := \begin{bmatrix} \frac{\sqrt{2}}{2} & \frac{\sqrt{3}}{3} & \frac{\sqrt{6}}{6} \\ \frac{\sqrt{2}}{2} & -\frac{\sqrt{3}}{3} & -\frac{\sqrt{6}}{6} \\ 0 & \frac{\sqrt{3}}{3} & -\frac{\sqrt{6}}{3} \end{bmatrix}$$

$$\mathit{true}$$

```
Apibrėžiame keitinio kintamuosius
> Y:=[y||(1..nops(X))];
```

$$Y := [\mathit{y1},\, \mathit{y2},\, \mathit{y3}]$$

ir apibrėžiame keitinį **K**=**PY**:

```
> evalm(X=P&*Y);
```

$$[x_1,\, x_2,\, x_3] = \left[\frac{\sqrt{2}\,\mathit{y1}}{2} + \frac{\sqrt{3}\,\mathit{y2}}{3} + \frac{\sqrt{6}\,\mathit{y3}}{6},\, \frac{\sqrt{2}\,\mathit{y1}}{2} - \frac{\sqrt{3}\,\mathit{y2}}{3} - \frac{\sqrt{6}\,\mathit{y3}}{6},\, \frac{\sqrt{3}\,\mathit{y2}}{3} - \frac{\sqrt{6}\,\mathit{y3}}{3}\right]$$

```
> K:=[seq(x[n]=rhs(%)[n],n=1..N)];
```

$$K := [x_1 = \frac{\sqrt{2}\,\mathit{y1}}{2} + \frac{\sqrt{3}\,\mathit{y2}}{3} + \frac{\sqrt{6}\,\mathit{y3}}{6},\, x_2 = \frac{\sqrt{2}\,\mathit{y1}}{2} - \frac{\sqrt{3}\,\mathit{y2}}{3} - \frac{\sqrt{6}\,\mathit{y3}}{6},\, x_3 = \frac{\sqrt{3}\,\mathit{y2}}{3} - \frac{\sqrt{6}\,\mathit{y3}}{3}]$$

Įsitikiname, kad šis keitinys kanonizuoja kvadratinę formą **Q**:

```
> simplify(subs(K,Q));
```

$$-2\,\mathit{y1}^2 + 3\,\mathit{y2}^2 + 6\,\mathit{y3}^2$$

Šitaip galime rasti teoriškai bet kokio kintamųjų skaičiaus kvadratin ės formos kanoninį pavidalą.

3. Raskime kvadratinės formos diagonalinį pavidalą Lagranžo metodu.

```
> restart:with(linalg):with(student):
```

Tarkime, kad mūsų kvadratinė forma tokia:

```
> Q:=3*x1^2+3*x2^2+3*x3^2-2*x1*x2-2*x1*x3+2*x2*x3;
```

$$Q := 3\,\mathit{x1}^2 + 3\,\mathit{x2}^2 + 3\,\mathit{x3}^2 - 2\,\mathit{x1}\,\mathit{x2} - 2\,\mathit{x1}\,\mathit{x3} + 2\,\mathit{x2}\,\mathit{x3}$$

completesquare komanda išskiriame kvadratinėje formoje pilną kvadratą kintamojo **x1** atžvilgiu:

```
> completesquare(Q,x1);
```

$$3\,(x1-\frac{x2}{3}-\frac{x3}{3})^2+\frac{8\,x2^2}{3}+\frac{4\,x2\,x3}{3}+\frac{8\,x3^2}{3}$$

Darome tą patį likusioje kvadratinės formos dalyje kinatmojo **x2** atžvilgiu:

```
> op(1,%)+completesquare(%-op(1,%),x2);
```

$$3\,(x1-\frac{x2}{3}-\frac{x3}{3})^2+\frac{8\,(x2+\frac{x3}{4})^2}{3}+\frac{5\,x3^2}{2}$$

Naujieji kvadratinės formos kintamieji - pilnųjų kvadratų pagrindai:

```
> indets(%,anything^2);
> {y1=op(1,%[1]),y2=op(1,%[2]),y3=op(1,%[3])};
> K:=solve(%,{x||(1..3)});
```

$$\{x3^2,\,(x2+\frac{x3}{4})^2,\,(x1-\frac{x2}{3}-\frac{x3}{3})^2\}$$

$$\{y1=x3,\;y2=x2+\frac{x3}{4},\;y3=x1-\frac{x2}{3}-\frac{x3}{3}\}$$

$$K:=\{x1=y3+\frac{y1}{4}+\frac{y2}{3},\;x2=y2-\frac{y1}{4},\;x3=y1\}$$

Pakeitę kintamuosius randame kanoninį kvadratinės formos pavidalą:

```
> subs(K,Q):simplify(%);
```

$$3\,y3^2+\frac{5\,y1^2}{2}+\frac{8\,y2^2}{3}$$

Pavyzdys formos, kurios rangas mažesnis už jos kintamųjų skaičių

```
> A3:=matrix(3,3,[1,2,3,2,0,2,3,2,5]);
```

$$A3:=\begin{bmatrix}1&2&3\\2&0&2\\3&2&5\end{bmatrix}$$

```
> X:=([x,y,z]);
```

$$X:=[x,\,y,\,z]$$

```
> Q3(X):=expand(evalm(transpose(X)&*A3&*X));
```

$$\mathrm{Q3}([x,\,y,\,z]):=x^2+4\,x\,y+6\,x\,z+4\,y\,z+5\,z^2$$

```
> DQ3:=jordan(A3,P);
```

$$DQ3:=\begin{bmatrix}0&0&0\\0&3-\sqrt{21}&0\\0&0&3+\sqrt{21}\end{bmatrix}$$

```
> print(P);
```

$$\begin{bmatrix} \frac{1}{3} & \frac{\sqrt{21}}{42}+\frac{1}{3} & -\frac{\sqrt{21}}{42}+\frac{1}{3} \\ \frac{1}{3} & -\frac{\sqrt{21}}{14}-\frac{1}{6} & \frac{\sqrt{21}}{14}-\frac{1}{6} \\ \frac{-1}{3} & -\frac{\sqrt{21}}{21}+\frac{1}{6} & \frac{\sqrt{21}}{21}+\frac{1}{6} \end{bmatrix}$$

Kvadratinės formos kanoninis pavidalas:

```
> X1:=([x1,y1,z1]);
```

$$X1 := [x1,\ y1,\ z1]$$

```
> multiply(transpose(X1),DQ3,X1);
```

$$y1^2\,(3-\sqrt{21}) + z1^2\,(3+\sqrt{21})$$

Matome, kad jos rangas lygus 2.

Uždaviniai.

1. Duota kvadratinė forma $3\,x^2 - 2\,y^2 + 15\,z^2 - 65\,xz$. Raskite jos kanoninį pavidalą, rangą, signatūrą. Užrašykite diagonalinę matricą ir ortogonalią diagonalizuojančią matricą. Patikrinkite jos ortogonalumą. Nustatykite kvadratinės formos apibr ėžtumą. Įsitikinkite diagonalizuojancios transformacijos pasirinkimo teisingumu

2. Tą patį padarykite su kvadratine forma
$Q := -4\,{x_1}^2 + 2\,{x_2}^2 - 5\,{x_3}^2 + 2\,x_1\,x_2 - 3\,x_1\,x_3 + 5\,x_2\,x_3$.

3. Sukonstruokite kvadratinę formą, kurios rangas mažesnis už kintamųjų skaičių. Nubrėžkite kreives (paviršius), kurių taškuose kvadratinė forma įgyja pastovias reikšmes.